

Campañas publicitarias – Parcial Lógico – 31/10/2009

Nos encargan hacer un programa para controlar las campañas¹ de varias agencias publicitarias y para ello tenemos la siguiente base de conocimiento.

Las campañas con

- el título
- el target o públicos a los que está dirigida la campaña
- los medios donde publicará la campaña y
- el anunciante

Los medios en donde pautan las agencias pueden tener la siguiente estructura:

- calle(calle, zona) define una campaña que tiene cartel en la calle. La zona puede ser una localidad de la provincia o bien un barrio (para Capital Federal)
- internet(sitio, cantidad de impresiones) define una campaña que se mostrará n cantidad de veces (impresiones) en un sitio de internet
- diario(diario, día, semanas) define una campaña que se mostrará en un diario y día determinado durante una x cantidad de semanas. El día de la semana es: lunes, martes, miércoles, etc.

campania(cocaColaMundial, [infantil, femenino, adulto], [calle(corrientes,centro), internet(lanacion,1000000)], cocaCola).

campania(barbiePrincesa, [infantil,femenino], [diario(clarin, domingo, 4)], mattel).

campania(lavarropaswirpool, [adulto, femenino], [calle(nuevedeJulio, lanus)], whirlpool).

campania(prestobarba, [adulto, adolescente, masculino], [], gillette).

El predicado agencia relaciona cada empresa con su agencia de publicidad:

agencia(cocaCola, latinmedios). ← primero el cliente, después la agencia

agencia(mattel, vivaAvisos).

agencia(whirlpool, vivaAvisos).

agencia(gillette, latinmedios).

Aclaración: una empresa sólo puede contratar los servicios de una agencia a la vez.

Las empresas se organizan en un ranking nacional según su importancia en el mercado:

ranking(cocaCola, 1).

ranking(mattel, 4).

ranking(whirlpool,15).

ranking(gillette,2).

Defina los predicados usando por lo menos una vez los siguientes conceptos

- Predicados de orden superior: al menos un uso de findall y un uso de forall
- Predicados generadores
- Recursividad
- Uso de explosión combinatoria
- Polimorfismo (específicamente, qué predicado se beneficia con este concepto)
- Los predicados 2, 3, 4, 5 y 6 deben ser inversibles

Indique en su solución dónde utilizó cada concepto.

La información que se agrega en el punto 1 puede ser usada en los siguientes puntos.

Puede utilizar los predicados sumlist/2 (suma una lista de números), member/2, length/2, además de forall/2, findall/3 y los operadores is, >, <, +, *, etc.

1-Agregue la información de cuánto cuesta publicar en cada medio teniendo en cuenta que:

¹ Una campaña publicitaria se define como “el conjunto de elementos y mensajes publicitarios que se realizan a través de los medios de comunicación “

- Publicar en una calle del centro sale \$10.000, en otros lugares sale \$5.000.
- Publicar en internet cuesta \$4 por cada impresión
- Publicar en un diario depende del día, si es domingo sale \$7.000, el resto de los días \$4.000 y a esto se debe multiplicar por la cantidad de semanas

2-**campaniaCara/1** una campaña es cara si todas sus publicaciones suma más de \$1.000.000 o se publica en la calle corrientes (del centro).

Ejemplo: ?- `campaniaCara(cocaColaMundial).`

Yes ← publica en calle corrientes (y además tiene un costo de \$4.010.000)

?- `campaniaCara(barbiePrincesa).`

No ← no publica en calle corrientes, y el costo de la campaña es de \$28.000

3-**campaniasPropicias/2** relaciona la lista de campañas que son propicias para un determinado público (es decir, si ese público forma parte del target).

Ejemplo: ?- `campaniasPropicias(Publico, Campanias).`

`Publico = infantil`

`Campanias = [cocaColaMundial, barbiePrincesa] ;`

`Publico = femenino`

`Campanias = [cocaColaMundial, barbiePrincesa, lavarropaswirpool]`

(...esto es sólo una muestra, hay más soluciones..)

4- **agenciaTop/2** relaciona las agencias que tienen al menos 2 clientes en los primeros n puestos del ranking.

Ejemplo: ?- `agenciaTop(Agencia, 15).`

`Agencia = latinmedios ;`

`Agencia = vivaAvisos ;`

(...)

5- **agenciaCheta/1** relaciona las agencias que solamente tienen como clientes a empresas que están en el top 10 en el ranking. *Por ejemplo:* latinmedios cumple esta condición (tiene de clientes a cocaCola que está primero y a gillette que está segundo).

Ejemplo: ?- `agenciaCheta(X).`

`X = latinmedios`

6- **facturacion/2** relaciona una agencia con el monto de su facturación (es la suma de los montos de las campañas de sus clientes).

Ejemplo: ?- `facturacion(Agencia, Total).`

`Agencia = latinmedios`

`Total = 4010000 ;` ← (todo de Coca Cola, gillette no puso avisos)

`Agencia = vivaAvisos`

`Total = 33000 ;` ← (\$28.000 de Mattel y \$5.000 de Whirlpool)

7- **avisosPosibles/3** relaciona los avisos que se pueden publicar con un monto de dinero determinado. La idea es contemplar todas las posibilidades que se pueden dar (incluyendo no hacer nada).

Ejemplo: ?- `avisosPosibles([calle(corrientes, centro), internet(lanacion, 100)], 500, Opciones).`

`Opciones = [internet(lanacion, 100)] ;`

`Opciones = []`

Con \$500 me alcanza para publicar el aviso en Internet (me cuesta \$400) o puedo no publicar nada.

Como no me alcanza para publicar el aviso en calle Corrientes del centro no forma parte de las soluciones posibles. **Nota:** asumir que los primeros dos argumentos siempre vienen instanciados (tienen valores).

Una solución posible

```
campania(cocaColaMundial, [infantil, femenino, adulto],
[calle(corrientes, centro), internet(lanacion, 1000000)], cocaCola).
campania(barbiePrincesa, [infantil, femenino], [diario(clarin,
domingo, 4)], mattel).
campania(lavarropaswhirlpool, [adulto, femenino], [calle(nueveDeJulio,
lanus)], whirlpool).
campania(prestobarba, [adulto, adolescente, masculino], [], gillette).
```

```
agencia(cocaCola, latinmedios).
agencia(mattel, vivaAvisos).
agencia(whirlpool, vivaAvisos).
agencia(gillette, latinmedios).
```

```
ranking(cocaCola, 1).
ranking(mattel, 4).
ranking(whirlpool, 15).
ranking(gillette, 2).
```

```
% punto 1 - cuanto cuesta publicar en un medio
cuantoCuesta(calle(_, centro), 10000).
cuantoCuesta(calle(_, Zona), 5000):-Zona \= centro.
cuantoCuesta(internet(_, Veces), Cuanto):- Cuanto is Veces * 4.
cuantoCuesta(diario(_, domingo, Semanas), Cuanto):-
    Cuanto is Semanas * 7000.
cuantoCuesta(diario(_, Dia, Semanas), Cuanto):-
    Dia \= domingo, Cuanto is Semanas * 4000.
% también podría usarse un predicado valorDia(Dia, Valor)
```

```
% punto 2 - determinar si una campaña es cara
% (más de 1.000.000 de pesos)
% o publicada en calle corrientes
campaniaCara(Campania):-
    costoCampania(Campania, Total), Total > 1000000.
campaniaCara(Campania):-
    publicadaEn(calle(corrientes, centro), Campania).
```

```
costoCampania(Campania, Total):-
    campania(Campania, _, Medios, _),
    costoMedios(Medios, Total).
```

```
costoMedios([], 0).
costoMedios([Medio|Medios], Total):-costoMedios(Medios, TotalMedios),
    cuantoCuesta(Medio, Costo),
    Total is Costo + TotalMedios.
costoMedios([Medio|Medios], Total):-costoMedios(Medios, TotalMedios),
    cuantoCuesta(Medio, Costo),
    Total is Costo + TotalMedios.
```

```
%Alternativa para costoMedios/2, usando un "map"
costoMedios(Medios, Total):-
    findall(Costo,
        (member(Medio, Medios), cuantoCuesta(Medio, Costo)),
        Costos),
    sumlist(Costos, Total).
```

```
publicadaEn(Calle, Campania):-
```

```

campania(Campania, _, Medios, _), member(Calle, Medios).

% punto 3 - relaciona la lista de campañas que son
% propicias para un determinado público
% (es decir, si ese público forma parte del target).
campaniasPropicias(Publico, Campanias):-publico(Publico),
    findall(Campania, campaniaPropiciaPara(Campania, Publico),
Campanias).

campaniaPropiciaPara(Campania, Publico):-
    campania(Campania, Target, _, _),
    member(Publico, Target).

publico(Publico):-campania(_, Target, _, _), member(Publico, Target).

% punto 4 - relaciona las agencias que tienen al menos
% 3 clientes en los primeros n puestos del ranking.
agenciaTop(Agencia, Posicion):- agencia(_, Agencia),
    findall(Cliente, clienteTop(Agencia, Cliente, Posicion), Clientes),
    length(Clientes, Cantidad), Cantidad >= 2.

clienteTop(Agencia, Cliente, Posicion):-
    agencia(Cliente, Agencia), ranking(Cliente, Ranking),
    Posicion >= Ranking.

% punto 5 - dice si una agencia tiene campañas solo de
% empresas que están en el top 10 en el ranking.
agenciaCheta(Agencia):-agencia(_, Agencia),
    forall(agencia(Cliente, Agencia), topTen(Cliente)).

topTen(Cliente):-ranking(Cliente, Ranking), Ranking < 11.

% punto 6 - relaciona una agencia con el monto de su facturación
% (es la suma de los montos de las campañas de sus clientes
facturacion(Agencia, Total):-agencia(_, Agencia),
    findall(Monto, (agencia(Cliente, Agencia),
        facturacionCliente(Cliente, Monto)), Montos),
    sumlist(Montos, Total).

facturacionCliente(Cliente, Monto):-
    findall(Costo, (campania(Campania, _, _, Cliente),
        costoCampania(Campania, Costo)), Total),
    sumlist(Total, Monto).

% punto 7 - relaciona los avisos que se pueden publicar
% con un monto de dinero determinado
avisosPosibles([], _, []).

avisosPosibles([Aviso|Avisos], Monto, [Aviso|AvisosPosibles]):-
    cuantoCuesta(Aviso, Costo), Monto >= Costo, Saldo is Monto - Costo,
    avisosPosibles(Avisos, Saldo, AvisosPosibles).

avisosPosibles([_|Avisos], Monto, AvisosPosibles):-
    avisosPosibles(Avisos, Monto, AvisosPosibles).

```